
 

 

 

 

 

 

Tipuri de date în PL/SQL 
 

 

Fiecare variabilă sau constantă utilizată într-un bloc PL/SQL este de un 

anumit tip prin care se specifică: 

 formatul său de stocare,  

 constrângerile care trebuie să le verifice 

 domeniul valorilor sale. 

Variabilele folosite în Oracle9i pot fi: 

 specifice PL/SQL; 

 nespecifice PL/SQL. 

Variabile specifice PL/SQL se clasifică în variabile: 

 de tip scalar,  

 compuse,  

 referinţă,  

 LOB (large objects), 

 tipuri obiect. 

Variabile nespecifice PL/SQL pot fi: 

 variabile de legătură (bind variables),  

 variabile gazdă (host variables),  

 variabile indicator. 

Variabile specifice PL/SQL 

Tipurile de date scalare 

Nu au componente interne (conţin valori atomice). Se împart în 5 clase. 

 Tipurile de date ce stochează valori numerice cuprind tipul NUMBER 

cu subtipurile DEC, DECIMAL, DOUBLE PRECISION, FLOAT, 

INTEGER, INT, NUMERIC, REAL, SMALLINT; tipul 

BINARY_INTEGER cu subtipurile NATURAL, NATURALN, 

POSITIVE, POSITIVEN, SIGNTYPE; tipul PLS_INTEGER. 

 Tipurile de date ce stochează caractere cuprind tipul VARCHAR2 cu 

subtipurile STRING, VARCHAR; tipul de date CHAR cu subtipul 

CHARACTER; tipurile LONG, RAW, LONG RAW, ROWID. 

 Tipurile de date ce stochează data calendaristică şi ora cuprind 

tipurile DATE, TIMESTAMP, TIMESTAMP WITH TIME ZONE, 

TIMESTAMP WITH LOCAL TIME ZONE, INTERVAL YEAR TO 

MONTH, INTERVAL DAY TO SECOND. 

 Tipurile de date globalizare ce stochează date unicode includ tipurile 

NCHAR şi NVARCHAR2.  

 Tipul de date BOOLEAN stochează valori logice (true, false sau null). 

Tipurile de date compuse  

Au componente interne care pot fi manipulate individual. Oracle oferă 

programatorului două tipuri de date compuse:  

 înregistrare (RECORD); 

 colecţie (INDEX-BY TABLE, NESTED TABLE, VARRAY).  

Tipurile de date referinţă  

REF CURSOR si REF obiect sunt tipuri de date ale căror valori, numite 

pointeri, fac referinţă către obiecte din program. Pointerii conţin locaţia de 

memorie (adresa) unui element şi nu elementul în sine. Tipul REF CURSOR este 

folosit pentru a face referinţă la un cursor explicit. Tipul REF obiect face 

referinţă la adresa unui obiect. 

Tipurile de date LOB  

Large object sunt acele tipuri de date ale căror valori, numite locatori 

(locators) specifică localizarea unor obiecte de dimensiuni mari, adică blocuri 

de date nestructurate, cum ar fi texte, imagini grafice, clipuri video şi sunete. 

Tipurile LOB sunt manipulate cu ajutorul pachetului DBMS_LOB.  

Aceste tipuri sunt:  

 CLOB (character large object),  

 BLOB (binary large object),  

 BFILE (binary file),  

 NCLOB (national language character large object).  

Tipurile obiect  

Sunt tipuri compuse, definite de utilizator, care încapsulează structuri de 

date (atribute) împreună cu subprograme pentru manipularea datelor (metode).  

Dintre tipurile scalare PL/SQL, următoarele sunt şi tipuri SQL (adică pot fi 

folosite pentru coloanele tabelelor Oracle): NUMBER, VARCHAR2, CHAR, 

LONG, RAW, LONG RAW, ROWID, NCHAR, NVARCHAR2, DATE. În unele 

cazuri, tipurile de date PL/SQL diferă de corespondentele lor SQL prin 

dimensiunea maximă permisă. 

Tipul NUMBER memorează numerele în virgulă fixă şi virgulă mobilă. El 

are forma generală NUMBER (m, n), unde m reprezintă numărul total de cifre, 

iar n numărul de zecimale. Valoarea unei variabile de tip NUMBER este 

cuprinsă între 1.0E-129 şi 9.99E125. Numărul de zecimale determină poziţia în 

care apare rotunjirea. Valoarea sa este cuprinsă între -84 şi 127, iar implicit este 

0. 

Tipul NUMBER are următoarele subtipuri, care au aceleaşi intervale de 

valori: NUMERIC, REAL, DEC, DECIMAL şi DOUBLE PRECISION (pentru 

memorarea datelor numerice în virgulă fixă), FLOAT (pentru memorarea datelor 

numerice în virgulă mobilă), SMALLINT, INTEGER şi INT (pentru memorarea 

numerelor întregi). Aceste subtipuri se pot utiliza pentru compatibilitate 

ANSI/ISO, IBM SQL/DS sau IBM DB2.  

Tipul BINARY_INTEGER memorează numere întregi cu semn având 

valori cuprinse între -231 - 1 şi 231 - 1. Acest tip de date este utilizat frecvent 

pentru indecşii tabelelor, nu necesită conversii şi admite mai multe subtipuri. De 

exemplu, pentru a restricţiona domeniul variabilelor la valori întregi nenegative 

se utilizează tipurile NATURAL (0 .. 231  – 1) şi POSITIVE (1 .. 231 – 1). 

Tipul PLS_INTEGER este utilizat pentru stocarea numerelor întregi cu 

semn şi are acelaşi interval de definire ca şi tipul BINARY_INTEGER. Operaţiile 

cu acest tip sunt efectuate mai rapid (folosesc aritmetica maşinii), decât cele cu 

tipurile NUMBER sau BINARY_INTEGER (folosesc librării aritmetice). Prin 

urmare, pentru o mai bună performanţă, este preferabil să se utilizeze tipul 

PLS_INTEGER. 

Variabilele alfanumerice pot fi de tip CHAR, VARCHAR2, LONG, RAW şi 

LONGRAW. Reprezentarea internă depinde de setul de caractere ales (ASCII sau 

EBCDIC).  

Tipurile CHAR, VARCHAR2 şi RAW pot avea un parametru pentru a 

preciza lungimea maximă. Dacă aceasta nu este precizată atunci, implicit, se 

consideră 1. Lungimea este exprimată în octeţi (nu în caractere). Subtipurile 

acestor tipuri se pot utiliza pentru compatibilitate ANSI/ISO, IBM SQL/DS sau 

IBM DB2.  

În Oracle9i a fost extinsă sintaxa pentru CHAR şi VARCHAR2, permiţând 

ca variabila ce precizează lungimea maximă să fie de tip CHAR sau BYTE. 

Variabilele de tip LONG pot memora texte, tabele de caractere sau 

documente, prin urmare şiruri de caractere de lungime variabilă de până la 

32760 octeţi. Este similar tipului VARCHAR2. 

Tipul RAW permite memorarea datelor binare (biţi) sau a şirurilor de 

octeţi. De exemplu, o variabilă RAW poate memora o secvenţă de caractere 

grafice sau o imagine digitizată. Tipul RAW este similar tipului alfanumeric, cu 

excepţia faptului că PL/SQL nu interpretează datele de tip RAW. Oracle nu face 

conversia datelor de acest tip, atunci când se transmit de la un sistem la altul. 

Chiar dacă lungimea maximă a unei variabile RAW poate fi 32767 octeţi, într-o 

coloană RAW a bazei de date nu se pot introduce decât 2000 octeţi. Pentru a 

insera valori mai mari se foloseşte o coloană de tip LONG RAW, care are 

lungimea maximă 231 octeţi. LONG RAW este similar tipului LONG, dar datele 

nu mai sunt interpretate de PL/SQL.  

Tipurile TIMESTAMP, TIMESTAMP WITH TIME ZONE, TIMESTAMP 

WITH LOCAL TIME ZONE, INTERVAL YEAR TO MONTH, INTERVAL DAY 

TO SECOND au fost introduse în Oracle9i şi permit rafinări ale tipului DATE. 

De exemplu, TIMESTAMP poate lua în considerare şi fracţiuni de secundă. 

PL/SQL suportă două seturi de caractere: una specifică bazei de date care 

este utilizată pentru definirea identificatorilor şi a codului sursă (database 

character set - DCS) şi o mulţime de caractere naţionale care este folosită pentru 

reprezentarea informaţiei cu caracter naţional (national character set - NCS). 

Tipurile de date NCHAR şi NVARCHAR2 sunt utilizate pentru stocarea în 

baza de date a şirurilor de caractere ce folosesc NCS. Ele oferă suport pentru 

globalizarea datelor, astfel încât utilizatorii din toată lumea pot interacţiona cu 

Oracle în limba lor naţională. Aceste tipuri de date suportă numai date Unicode.  

Unicode este o mulţime de caractere globale care permite stocarea de 
informaţie în orice limbă, folosind o mulţime unică de caractere.  

Prin urmare, unicode furnizează o valoare cod unică pentru fiecare 
caracter, indiferent de platformă, program sau limbă. 

Variabile nespecifice PL/SQL 

Variabila de legătură (bind) se declară într-un mediu gazdă şi este folosită 
pentru transferul (la execuţie) valorilor numerice sau de tip caracter în/din unul 
sau mai multe programe PL/SQL. Variabilele declarate în mediul gazdă sau în 

cel apelant pot fi referite în instrucţiuni PL/SQL, dacă acestea nu sunt în cadrul 
unei proceduri, funcţii sau pachet.  

În SQL*Plus, variabilele de legătură se declară folosind comanda 
VARIABLE, iar pentru afişarea valorilor acestora se utilizează comanda PRINT. 
Ele sunt referite prin prefixarea cu simbolul „:“, pentru a putea fi deosebite de 
variabilele declarate în PL/SQL. 

Deoarece instrucţiunile SQL pot fi integrate în programe C, este necesar 
un mecanism pentru a transfera valori între mediul de programare C şi 

instrucţiunile SQL care comunică cu server-ul bazei de date Oracle. În acest 
scop, în programul încapsulat sunt definite variabilele gazdă (host). Acestea sunt 
declarate între directivele BEGIN DECLARE SECTION şi END DECLARE 
SECTION ale preprocesorului. 

O valoare null în baza de date nu are o valoare corespunzătoare în mediul 
limbajului gazdă (de exemplu, limbajul C). Pentru a rezolva problema 
comunicării valorilor null între programul scris în limbaj gazdă şi sistemul 
Oracle, au fost definite variabilele indicator. Acestea sunt variabile speciale de 

tip întreg, folosite pentru a indica dacă o valoare null este recuperată (extrasă) 
din baza de date sau stocată în aceasta. Ele au următoarea formă: 

:nume_extern [: indicator] 

De exemplu, dacă atribuirea este făcută de limbajul gazdă, valoarea –1 a 
indicatorului specifică faptul că PL/SQL trebuie să înlocuiască valoarea 
variabilei prin null, iar o valoare a indicatorului mai mare ca zero precizează că 
PL/SQL trebuie să considere chiar valoarea variabilei. 

Declararea variabilelor 

Identificatorii PL/SQL trebuie declaraţi înainte de a fi referiţi în blocul 
PL/SQL. Dacă în declaraţia unei variabile apar referiri la alte variabile, acestea 
trebuie să fi fost declarate anterior. Orice variabilă declarată într-un bloc este 
accesibilă blocurilor conţinute sintactic în acesta. 

Tipurile scalare sunt predefinite în pachetul STANDARD. Pentru a folosi 
un astfel de tip într-un program este suficient să fie declarată o variabilă de tipul 

respectiv.  

Tipurile compuse sunt definite de utilizator. Prin urmare, în acest caz 

trebuie definit efectiv tipul şi apoi declarată variabila de tipul respectiv. 

În declararea variabilelor pot fi utilizate atributele %TYPE şi 

%ROWTYPE, care reprezintă tipuri de date implicite. Aceste tipuri permit 

declararea unei variabile în concordanţă cu declaraţii de variabile făcute 

anterior.  

Atributul %TYPE permite definirea unei variabile având tipul unei 

variabile declarate anterior sau tipul unei coloane dintr-un tabel.  

Atributul %ROWTYPE permite definirea unei variabile având tipul unei 

înregistrări dintr-un tabel. Avantajul utilizării acestui atribut constă în faptul că 

nu este necesar să se cunoască numărul şi tipurile coloanelor tabelului. 

Elementele individuale ale acestei structuri de tip înregistrare sunt referite în 

maniera clasică, prefixând numele coloanei cu numele variabilei declarate.  

Calitatea atributelor %TYPE şi %ROWTYPE constă în faptul că simplifică 

întreţinerea codului PL/SQL. De exemplu, poate fi modificată dimensiunea unei 

coloane, fără să fie necesară modificarea declaraţiei variabilelor al căror tip s-a 

definit făcând referinţă la tipul coloanei respective. 

Sintaxa declarării unei variabile este următoarea: 

identificator [CONSTANT] {tip_de_date | identificator%TYPE | 

identificator%ROWTYPE} [NOT NULL] 

[ {:= | DEFAULT} expresie_PL/SQL]; 

Se pot defini constante (valoarea stocată nu poate fi modificată) prin 

specificarea la declarare a cuvântului cheie CONSTANT.  

Exemplu: 

v_valoare        NUMBER(15) NOT NULL := 0; 

v_data_achizitie DATE DEFAULT SYSDATE; 

v_material       VARCHAR2(15) := 'Matase'; 

c_valoare        CONSTANT NUMBER := 100000; 

v_stare          VARCHAR2(20) DEFAULT 'Buna'; 

v_clasificare    BOOLEAN  DEFAULT FALSE; 

v_cod_opera      opera.cod_operaTYPE; 

v_opera          operaROWTYPE; 
int_an_luna      INTERVAL YEAR TO MONTH :=  

                   INTERVAL '3-2' YEAR TO MONTH; 

Observaţii: 

1.  Pentru a denumi o variabilă este utilizată frecvent (pentru uşurinţa referirii) 

prefixarea cu litera v (v_identificator), iar pentru o constantă este folosită 

prefixarea cu litera c (c_identificator). 



 7 

2.  Variabilele pot fi iniţializate, iar dacă o variabilă nu este iniţializată, valoarea 

implicită a acesteia este null. Dacă o variabilă este declarată NOT NULL, 

atunci ea va fi obligatoriu iniţializată. 

3.  Pentru a iniţializa o variabilă sau o constantă poate fi utilizată o expresie 

PL/SQL compatibilă ca tip cu variabila sau constanta respectivă. 

4.  Constantele trebuie iniţializate când sunt declarate, altfel apare o eroare la 

compilare.  

5.  În secţiunea declarativă, pe fiecare linie, există o singură declaraţie de 

variabilă. 

6.  Două obiecte (variabile) pot avea acelaşi nume cu condiţia să fie definite în 

blocuri diferite. Dacă ele coexistă, poate fi folosit doar obiectul declarat în 

blocul curent. 

7.  Atributul %ROWTYPE nu poate include clauze de iniţializare. 

Definirea subtipurilor 

Subtipurile derivă dintr-un tip de bază, la care se adaugă anumite restricţii. 

De exemplu, NATURAL este un subtip predefinit PL/SQL, derivat din tipul de 

bază BINARY_INTEGER, cu restricţia că permite prelucrarea valorilor întregi 

nenegative.  

Prin urmare, un subtip nu reprezintă un nou tip de date, ci un tip existent 

asupra căruia se aplică anumite constrângeri. Subtipurile presupun acelaşi set de 

operaţii ca şi tipul de bază, dar aplicate unui subset de valori al acestui tip. 

Sistemul Oracle permite ca utilizatorul să-şi definească propriile sale 

tipuri şi subtipuri de date în partea declarativă a unui bloc PL/SQL, subprogram 

sau pachet utilizând sintaxa: 

SUBTYPE nume_subtip IS tip_de_baza [NOT NULL]; 

În dicţionarul datelor există vizualizări care furnizează informaţii despre 

tipurile de date create de utilizator (USER_TYPES, USER_TYPE_ATTRS). 

Conversii între tipuri de date 

Există două tipuri de conversii:  

implicite; 

explicite.  

PL/SQL face automat conversii implicite între caractere şi numere sau 

între caractere şi date calendaristice. Chiar dacă sistemul realizează automat 

aceste conversii, în practică se utilizează frecvent funcţii de conversie explicită.  

Funcţiile de conversie explicită din SQL sunt utilizabile şi în PL/SQL. 

Acestea sunt: TO_NUMBER, TO_CHAR, TO_DATE, TO_MULTI_BYTE, 

TO_SINGLE_BYTE, CHARTOROWID, ROWIDTOCHAR, RAWTOHEX, 

HEXTORAW, TO_CLOB, TO_LOB. 

În Oracle9i se pot folosi următoarele funcţii de conversie: ASCIISTR, 

BIN_TO_NUM, NUMTODSINTERVAL, TO_TIMESTAMP, TO_YMINTERVAL, 

TO_NCHAR, TO_NCLOB, TO_TIMESTAMP_TZ, NUMTOYMINTERVAL, 

TO_DSINTERVAL, REFTOHEX, RAWTOHEX, RAWTONHEX, FROM_TZ, 

ROWIDTONCHAR, COMPOSE, DECOMPOSE. 
Denumirile acestor funcţii reflectă posibilităţile pe care le oferă. De 

exemplu, TO_YMINTERVAL converteşte argumentele sale la tipul INTERVAL 
YEAR TO MONTH conform unui format specificat. Funcţia COMPOSE 
converteşte un şir de caractere la un şir unicode (asociază o valoare cod unică 
pentru fiecare simbol din şir). 

Înregistrări 

Tipul RECORD oferă un mecanism pentru prelucrarea înregistrărilor. 
Înregistrările au mai multe câmpuri ce pot fi de tipuri diferite, dar care sunt 
legate din punct de vedere logic.  

Inregistrările trebuie definite în doi paşi:  

 se defineşte tipul RECORD; 

 se declară înregistrările de acest tip. 

Declararea tipului RECORD se face conform următoarei sintaxe: 

TYPE nume_tip  IS RECORD 
(nume_câmp1 {tip_câmp | variabilă%TYPE | 
  nume_tabel.coloană%TYPE | nume_tabel%ROWTYPE} 
  [ [NOT NULL] {:= | DEFAULT} expresie1], 
(nume_câmp2 {tip_câmp | variabilă%TYPE | 
  nume_tabel.coloană%TYPE | nume_tabel%ROWTYPE} 
  [ [NOT NULL] {:= | DEFAULT} expresie2],…); 

Identificatorul nume_tip reprezintă numele tipului RECORD care se va 
specifica în declararea înregistrărilor, nume_câmp este numele unui câmp al 
înregistrării, iar tip_câmp este tipul de date al câmpului. 

Observaţii: 

 Dacă un câmp nu este iniţializat atunci implicit se consideră că are 
valoarea NULL. Dacă s-a specificat constrângerea NOT NULL, atunci 
obligatoriu câmpul trebuie iniţializat cu o valoare diferită de NULL. 

 Pentru referirea câmpurilor individuale din înregistrare se prefixează 
numele câmpului cu numele înregistrării. 

 Pot fi asignate valori unei înregistrări utilizând comenzile SELECT, 
FETCH sau instrucţiunea clasică de atribuire. De asemenea, o înregistrare 
poate fi asignată altei înregistrari de acelaşi tip. 

 Componentele unei înregistrări pot fi de tip scalar, RECORD, TABLE, 
obiect, colecţie (dar, nu tipul REF CURSOR). 

 PL/SQL permite declararea şi referirea înregistrărilor imbricate. 

 Numărul de câmpuri ale unei înregistrări nu este limitat. 

 Înregistrările nu pot fi comparate (egalitate, inegalitate sau null). 

 Înregistrările pot fi parametri în subprograme şi pot să apară în clauza 
RETURN a unei funcţii. 

Diferenţa dintre atributul %ROWTYPE şi tipul de date compus RECORD: 

 tipul RECORD permite specificarea tipului de date pentru câmpuri şi 
permite declararea câmpurilor sale; 

 atributul %ROWTYPE nu cere cunoaşterea numărului şi tipurilor 
coloanelor tabloului.  

Oracle9i introduce câteva facilităţi legate de acest tip de date. 

 Se poate insera (INSERT) o linie într-un tabel utilizând o înregistrare. 
Nu mai este necesară listarea câmpurilor individuale, ci este suficientă 
utilizarea numelui înregistrării. 

 Se poate reactualiza (UPDATE) o linie a unui tabel utilizând o 
înregistrare. Sintaxa SET ROW permite să se reactualizeze întreaga 
linie folosind conţinutul unei înregistrări. 

 Într-o înregistrare se poate regăsi şi returna sau şterge informaţia din 
clauza RETURNING a comenzilor UPDATE sau DELETE.  

 Dacă în comenzile UPDATE sau DELETE se modifică mai multe linii, 
atunci pot fi utilizate în sintaxa BULK COLLECT INTO, colecţii de 
înregistrări. 

Exemplu: 
Exemplul următor arată modul în care poate să fie utilizată o înregistrare 

în clauza RETURNING asociată comenzii DELETE. 

DECLARE 

  TYPE val_opera IS RECORD ( 

    cheie  NUMBER, 

    val    NUMBER); 

  v_info_valoare  val_opera; 

BEGIN 

  DELETE FROM opera 

      WHERE cod_opera = 753 

    RETURNING cod_opera, valoare 

      INTO v_info_valoare; 

  … 

END; 

Colecţii 

Uneori este preferabil să fie prelucrate simultan mai multe variabile de 
acelaşi tip. Tipurile de date care permit acest lucru sunt colecţiile. Fiecare 
element are un indice unic, care determină poziţia sa în colecţie.  

Oracle7 a furnizat tipul index-by table, iniţial numit PL/SQL table datorită 
asemănării sale cu structura tabelelor relaţionale.  

Oracle8 a introdus două tipuri colecţie, nested table şi varray. Oracle9i 
permite crearea de colecţii pe mai multe niveluri, adică colecţii de colecţii.  

În PL/SQL există trei tipuri de colecţii:  

 tablouri indexate (index-by tables); 

 tablouri imbricate (nested tables); 

 vectori (varrays sau varying arrays). 

Tipul index-by table poate fi utilizat numai în declaraţii PL/SQL. Tipurile 
varray şi nested table pot fi utilizate atât în declaraţii PL/SQL, cât şi în declaraţii 
la nivelul schemei (de exemplu, pentru definirea tipului unei coloane a unui 
tabel relaţional).  

Exemplu: 
În exemplul care urmează sunt ilustrate cele trei tipuri de colecţii. 

DECLARE  

  TYPE  tab_index IS TABLE OF NUMBER 

    INDEX BY BINARY_INTEGER; 

  TYPE  tab_imbri IS TABLE OF NUMBER; 

  TYPE  vector IS VARRAY(15) OF NUMBER; 

  v_tab_index  tab_index; 

  v_tab_imbri  tab_imbri; 

  v_vector     vector; 

BEGIN 

  v_tab_index(1) := 72; 

  v_tab_index(2) := 23; 

  v_tab_imbri := tab_imbri(5, 3, 2, 8, 7); 

  v_vector := vector(1, 2); 

END; 

Observaţii: 

 Deoarece colecţiile nu pot fi comparate (egalitate sau inegalitate), ele 
nu pot să apară în clauzele DISTINCT, GROUP BY, ORDER BY.  

 Tipul colecţie poate fi definit într-un pachet. 

 Tipul colecţie poate să apară în clauza RETURN a unei funcţii.  

 Colecţiile pot fi parametri formali într-un subprogram. 

 Accesul la elementele individuale ale unei colecţii se face prin 
utilizarea unui indice.  

Tablouri indexate 

Tipul de date index-by table oferă un mecanism pentru prelucrarea 

tablourilor. Tabloul indexat PL/SQL are două componente: o coloană ce 

cuprinde cheia primară pentru acces la liniile tabloului şi o coloană care include 

valoarea efectivă a elementelor tabloului.  

Oracle7 asigură definirea tablourilor de înregistrări care pot fi declarate şi 

utilizate numai în programe PL/SQL, Oracle8 realizează definirea tablourilor de 

tipuri obiect, iar Oracle9i permite definirea tablourilor de colecţii. 

În Oracle9i tipul index-by table este redenumit associative array pentru 

compatibilitate (de limbaj) cu termenul folosit în alte limbaje de programare 

(C++, JavaScript, Perl) pentru a defini această structură de date.  

Tablourile indexate PL/SQL trebuie definite în doi paşi: se defineşte tipul 

TABLE; se declară tabloul indexat PL/SQL de acest tip. 

Declararea tipului TABLE se face respectând următoarea sintaxă: 

TYPE nume_tip  IS TABLE OF  

{tip_coloană | variabilă%TYPE |  

nume_tabel.coloană%TYPE [NOT NULL] |  

nume_tabel%ROWTYPE}  

INDEX BY tip_indexare; 

Identificatorul nume_tip este numele noului tip definit care va fi specificat 

în declararea tabloului PL/SQL, iar tip_coloană este un tip scalar simplu (de 

exemplu, VARCHAR2, CHAR, DATE sau NUMBER).  

Până la versiunea Oracle9i unicul tip de indexare acceptat era INDEX BY 

BINARY_INTEGER. Oracle9i permite următoarele opţiuni pentru tip_indexare: 

PLS_INTEGER, NATURAL, POSITIVE, VARCHAR2(n) sau chiar indexarea 

după un tip declarat cu %TYPE. Nu sunt permise indexările INDEX BY 

NUMBER, INDEX BY INTEGER, INDEX BY DATE, INDEX BY VARCHAR2, 

INDEX BY CHAR(n) sau indexarea după un tip declarat cu %TYPE în care 

intervine unul dintre tipurile enumerate anterior.  

Observaţii: 

 Elementele unui tablou indexat nu sunt într-o ordine particulară şi pot 

fi inserate cu chei arbitrare.  

 Deoarece nu există constrângeri de dimensiune, dimensiunea tabloului 

se modifică dinamic. 

 Tabloul indexat PL/SQL nu poate fi iniţializat în declararea sa.  

 Un tablou indexat neiniţializat este vid (nu conţine nici valori, nici 

chei). 

 Un element al tabloului este nedefinit atâta timp cât nu are atribuită o 
valoare efectivă.  

 Iniţial, un tablou indexat este nedens. După declararea unui tablou se 
poate face referire la liniile lui prin precizarea valorii cheii primare. 

 Dacă se face referire la o linie care nu există, atunci se produce 
excepţia NO_DATA_FOUND.  

 Dacă se doreşte contorizarea numărului de linii, trebuie declarată o 
variabilă în acest scop sau poate fi utilizată o metodă asociată 
tabloului.  

 Deoarece numărul de linii nu este limitat, operaţia de adăugare de linii 
este restricţionată doar de dimensiunea memoriei alocate. 

 Tablourile pot să apară ca argumente într-o procedură. 

Pentru inserarea unor valori din tablourile PL/SQL într-o coloană a unui 
tabel de date se utilizează instrucţiunea INSERT în cadrul unei secvenţe 
repetitive LOOP.  

Asemănător, pentru regăsirea unor valori dintr-o coloană a unei baze de 
date într-un tablou PL/SQL se utilizează instrucţiunea FETCH (cursoare) sau 
instrucţiunea de atribuire în cadrul unei secvenţe repetitive LOOP. 

Pentru a şterge liniile unui tablou fie se asignează elementelor tabloului 
valoarea null, fie se declară un alt tablou PL/SQL (de acelaşi tip) care nu este 
iniţializat şi acest tablou vid se asignează tabloului PL/SQL care trebuie şters. În 
PL/SQL 2.3 ştergerea liniilor unui tabel se poate face utilizând metoda DELETE. 

Exemplu: 
Să se definească un tablou indexat PL/SQL având elemente de tipul 

NUMBER. Să se introducă 20 de elemente în acest tablou. Să se şteargă tabloul. 

DECLARE  

  TYPE  tablou_numar IS TABLE OF NUMBER 

    INDEX BY PLS_INTEGER; 

  v_tablou  tablou_numar; 

BEGIN 

  FOR i IN 1..20 LOOP 

    v_tablou(i) := i*i;  

    DBMS_OUTPUT.PUT_LINE(v_tablou(i)); 

  END LOOP; 

  --v_tablou := NULL; 

  --aceasta atribuire da eroarea PLS-00382 

  FOR i IN v_tablou.FIRST..v_tablou.LAST LOOP 

    v_tablou(i) := NULL;  

  END LOOP; 

  DBMS_OUTPUT.PUT_LINE('tabloul are ' || v_tablou.COUNT || 

                       ' elemente'); 

END; 

În PL/SQL este folosit frecvent tipul tablou de înregistrări. Referirea la un 
element al tabloului se face prin forma clasică: tabel(index).câmp. 

Exemplu: 
 Să se definească un tablou de înregistrări având tipul celor din tabelul 
organizator. Să se iniţializeze un element al tabloului şi să se introducă în 
tabelul organizator. Să se şteargă elementele tabloului. 

DECLARE 

  TYPE org_table_type IS TABLE OF organizator%ROWTYPE 

    INDEX BY BINARY INTEGER; 

  org_table  org_table_type; 

  i          NUMBER; 

BEGIN 

  IF org_table.COUNT <>0 THEN 

    i := org_table.LAST+1; 

    ELSE i:=1; 

  END IF; 

  org_table(i).cod_org := 752; 

  org_table(i).nume := 'Grigore Ion'; 

  org_table(i).adresa := 'Calea Plevnei 18 Sibiu'; 

  org_table(i).tip := 'persoana fizica'; 

  INSERT INTO organizator 

  VALUES (org_table(i).cod_org, org_table(i).nume, 

          org_table(i).adresa, org_table(i).tip); 

  -- sau folosind noua facilitate Oracle9i 

  -- INSERT INTO organizator 

  -- VALUES (org_table(i)); 

  org_table.DELETE; -- sterge toate elementele 

  DBMS_OUTPUT.PUT_LINE('Dupa aplicarea metodei DELETE  

      sunt '||TO_CHAR(org_table.COUNT)||' elemente'); 

END; 

Vectori 

Vectorii (varray) sunt structuri asemănătoare vectorilor din limbajele C 
sau Java. Spre deosebire de tablourile indexate, vectorii au o dimensiune 
maximă (constantă) stabilită la declarare. În special, se utilizează pentru 
modelarea relaţiilor one-to-many, atunci când numărul maxim de elemente din 
partea „many“ este cunoscut şi ordinea elementelor este importantă.  

Vectorii reprezintă structuri dense. Fiecare element are un index care dă 
poziţia sa în vector şi care este folosit pentru accesarea elementelor particulare. 
Limita inferioară a indicelui este 1. Vectorul poate conţine un număr variabil de 
elemente, de la 0 (vid) la numărul maxim specificat obligatoriu în definiţia sa.  

Tipul de date vector este declarat utilizând sintaxa: 

TYPE nume_tip IS  
{VARRAY | VARYING ARRAY} (lungime_maximă) 
OF tip_elemente [NOT NULL]; 

Identificatorul nume_tip este numele tipului de date vector, iar 
lungime_maximă reprezintă numărul maxim de elemente din vector. 
Tip_elemente este un tip scalar PL/SQL, tip înregistrare sau tip obiect. De 
asemenea, acest tip poate fi definit utilizând atributele %TYPE sau 
%ROWTYPE.  

În Oracle9i sunt permise (pentru tip_elemente) tipurile TABLE sau alt tip 
VARRAY. Există restricţii referitoare la tipul elementelor, în sensul că acesta nu 
poate să fie BOOLEAN, NCHAR, NCLOB, NVARCHAR2, REF CURSOR, 
PLS_INTEGER, LONG, LONG RAW, NATURAL, NATURALN, POSITIVE, 
POSITIVEN, BINARY_INTEGER, SIGNTYPE, STRING, tip obiect cu atribute 
TABLE sau VARRAY, BLOB, CLOB, tip obiect cu atribute BLOB sau CLOB. 

Exemplu: 

DECLARE 

  TYPE secventa IS VARRAY(5) OF VARCHAR2(10); 

  v_sec  secventa := secventa ('alb', 'negru', 'rosu',  

                               'verde'); 

BEGIN 

  v_sec (3) := 'rosu'; 

  v_sec.EXTEND; -- adauga un element null 

  v_sec(5) := 'albastru'; 

  -- extinderea la 6 elemente va genera eroarea ORA-06532 

  v_sec.EXTEND; 

END; 

Tablouri imbricate 

Tablourile imbricate (nested table) sunt tablouri indexate a căror 
dimensiune nu este stabilită. Numărul maxim de linii ale unui tablou imbricat 
este dat de capacitatea maximă 2 GB. 

Un tablou imbricat este o mulţime neordonată de elemente de acelaşi tip. 
Valorile de acest tip: 

-- pot fi stocate în baza de date,  
-- pot fi prelucrate direct în instrucţiuni SQL  
-- au excepţii predefinite proprii.  

Sistemul Oracle nu stochează liniile unui tablou imbricat într-o ordine 

particulară. Dar, când se regăseşte tabloul în variabile PL/SQL, liniile vor avea 

indici consecutivi începând cu valoarea 1. Iniţial, aceste tablouri sunt structuri 

dense, dar se poate ca în urma prelucrării să nu mai aibă indici consecutivi. 

Comanda de declarare a tipului de date tablou imbricat are sintaxa: 

TYPE nume_tip IS TABLE OF tip_ elemente [NOT NULL]; 

Identificatorul nume_tip reprezintă numele noului tip de date tablou 

imbricat, iar tip_elemente este tipul fiecărui element din tabloul imbricat, care 

poate fi un tip definit de utilizator sau o expresie cu %TYPE, respectiv 

%ROWTYPE.  

În Oracle9i sunt permise (pentru tip_elemente) tipurile TABLE sau alt tip 

VARRAY. Există restricţii referitoare la tipul elementelor, în sensul că acesta nu 

poate să fie BOOLEAN, STRING, NCHAR, NCLOB, NVARCHAR2, REF 

CURSOR, BINARY_INTEGER, PLS_INTEGER, LONG, LONG RAW, 

NATURAL, NATURALN, POSITIVE, POSITIVEN, SIGNTYPE, tip obiect cu 

atributele TABLE sau VARRAY. 

Tabloul imbricat are o singură coloană, iar dacă aceasta este de tip obiect, 

tabloul poate fi vizualizat ca un tabel multicoloană, având câte o coloană pentru 

fiecare atribut al tipului obiect. 

Exemplu: 

DECLARE 

  TYPE numartab IS TABLE OF NUMBER; 

  -- se creeaza un tablou cu un singur element 

  v_tab_1  numartab := numartab(-7); 

  -- se creeaza un tablou cu 4 elemente 

  v_tab_2  numartab := numartab(7,9,4,5); 

  -- se creeaza un tablou fara nici un element 

  v_tab_3  numartab := numartab(); 

BEGIN 

  v_tab_1(1) := 57; 

  FOR j IN 1..4 LOOP 

    DBMS_OUTPUT.PUT_LINE (v_tab_2(j) || ' '); 

  END LOOP; 

END; 

Se observă că singura diferenţă sintactică între tablourile indexate şi cele 

imbricate este absenţa clauzei INDEX BY BINARY_INTEGER. Mai exact, dacă 

această clauză lipseşte, tipul este tablou imbricat. 

Observaţii: 

 Spre deosebire de tablourile indexate, vectorii şi tablourile imbricate 

pot să apară în definirea tabelelor bazei de date. 

 Tablourile indexate pot avea indice negativ, domeniul permis pentru 

index fiind –2147483647..2147483647, iar pentru tabele imbricate 

domeniul indexului este 1..2147483647. 

 Tablourile imbricate, spre deosebire de tablourile indexate, pot fi 

prelucrate prin comenzi SQL. 

 Tablourile imbricate trebuie iniţializate şi/sau extinse pentru a li se 

adăuga elemente. 

Când este creat un tablou indexat care nu are încă elemente, el este vid. 

Dacă un tablou imbricat (sau un vector) este declarat, dar nu are încă nici un 

element (nu este iniţializat), el este automat iniţializat (atomic) null. Adică, 

colecţia este null, nu elementele sale. Prin urmare, pentru tablouri imbricate 

poate fi utilizat operatorul IS NULL. Dacă se încearcă să se adauge un element 

la un tablou imbricat null, se va genera eroarea „ORA - 06531: reference to 

uninitialized collection“ care corespunde excepţiei predefinite 

COLLECTION_IS_NULL. 

Prin urmare, cum poate fi iniţializat un tablou imbricat? Ca şi obiectele, 

vectorii şi tablourile imbricate sunt iniţializate cu ajutorul constructorului. 

Acesta are acelaşi nume ca şi tipul colecţiei referite. PL/SQL apelează un 

constructor numai în mod explicit. Tabelele indexate nu au constructori. 

Constructorul primeşte ca argumente o listă de valori de tip tip_elemente. 

Elementele sunt numerotate în ordine, de la 1 la numărul de valori date ca 

parametrii constructorului. Dimensiunea iniţială a colecţiei este egală cu 

numărul de argumente date în constructor, când aceasta este iniţializată. Pentru 

vectori nu poate fi depăşită dimensiunea maximă precizată la declarare. Atunci 

când constructorul este fără argumente, va crea o colectie fără nici un element 

(vida), dar care are valoarea not null. Exemplul următor este concludent în acest 

sens. 

Exemplu: 

DECLARE 

  TYPE  alfa IS TABLE OF VARCHAR2(50); 

  -- creeaza un tablou (atomic) null 

  tab1  alfa ; 

  /* creeaza un tablou cu un element care este null, dar 

     tabloul nu este null, el este initializat, poate 

     primi elemente */ 

  tab2  alfa := alfa() ; 

BEGIN 

 

  IF tab1 IS NULL THEN 

    DBMS_OUTPUT.PUT_LINE('tab1 este NULL'); 

  ELSE 

    DBMS_OUTPUT.PUT_LINE('tab1 este NOT NULL'); 

  END IF; 

  IF tab2 IS NULL THEN 

    DBMS_OUTPUT.PUT_LINE('tab2 este NULL'); 

  ELSE 

    DBMS_OUTPUT.PUT_LINE('tab2 este NOT NULL'); 

  END IF; 

END; 

În urma execuţiei acestui bloc se obţine următorul rezultat: 

tab1 este NULL 

tab2 este NOT NULL 

Excepţiile semnificative care apar în cazul utilizării incorecte a colecţiilor: 

Exemplu: 

DECLARE 

  TYPE  numar IS TABLE OF INTEGER; 

  alfa  numar; 

BEGIN 

  alfa(1) := 77; 

  -- declanseaza exceptia COLLECTION_IS_NULL 

  alfa := numar(15, 26, 37); 

  alfa(1) := ASCII('X'); 

  alfa(2) := 10*alfa(1); 

  alfa('P') := 77; 

  /* declanseaza exceptia VALUE_ERROR deoarece indicele 

     nu este convertibil la intreg */ 

  alfa(4) := 47; 

  /* declanseaza exceptia SUBSCRIPT_BEYOND_COUNT deoarece 

     indicele se refera la un element neinitializat */ 

  alfa(null) := 7; -- declanseaza exceptia VALUE_ERROR 

  alfa(0) := 7; -- exceptia SUBSCRIPT_OUTSIDE_LIMIT 

  alfa.DELETE(1); 

  IF alfa(1) = 1 THEN … -- exceptia NO_DATA_FOUND 

  … 

END; 

Tablourile imbricate şi vectorii pot fi utilizaţi drept câmpuri în tabelele 

bazei. Aceasta presupune că fiecare înregistrare din tabelul respectiv conţine un 

obiect de tip colecţie. Înainte de utilizare, tipul trebuie stocat în dicţionarul 

datelor, deci trebuie declarat prin comanda: 

CREATE TYPE nume_tip AS {TABLE | VARRAY} OF tip_elemente; 

După crearea tabelului (prin comanda CREATE TABLE), pentru fiecare 

câmp de tip tablou imbricat din tabel este necesară clauza de stocare:  

NESTED TABLE nume_câmp STORE AS nume_tabel; 

Colecţii pe mai multe niveluri 

În Oracle9i se pot construi colecţii pe mai multe niveluri (multilevel 

collections), prin urmare colecţii ale căror elemente sunt, în mod direct sau 

indirect, colecţii. În felul acesta pot fi definite structuri complexe: vectori de 

vectori, vectori de tablouri imbricate, tablou imbricat de vectori, tablou imbricat 

de tablouri imbricate, tablou imbricat sau vector de un tip definit de utilizator 

care are un atribut de tip tablou imbricat sau vector.  

Aceste structuri complexe pot fi utilizate ca tipuri de date pentru definirea: 

 coloanelor unui tabel relaţional,  

 atributelor unui obiect într-un tabel obiect, 

 variabilelor PL/SQL.  

Observaţii: 

 Numărul nivelurilor de imbricare este limitat doar de capacitatea de 

stocare a sistemului. 

 Pentru a accesa un element al colecţiei incluse sunt utilizate două 

seturi de paranteze.  

 Obiectele de tipul colecţie pe mai multe niveluri nu pot fi comparate. 

Exemplu: 

În exemplele care urmează sunt definite trei structuri complexe şi sunt 

prezentate câteva modalităţi de utilizare ale acestora. Exemplele se referă la 

vectori pe mai multe niveluri, tablouri imbricate pe mai multe niveluri şi tablouri 

indexate pe mai multe niveluri. 

DECLARE  

  TYPE  alfa IS VARRAY(10) OF INTEGER; 

  TYPE  beta IS VARRAY(10) OF alfa; 

  valf  alfa := alfa(12,31,5); --initializare 

  vbet  beta := beta(valf,alfa(55,6,77),alfa(2,4),valf); 

  i     integer; 

  var1  alfa; 

BEGIN 

  i := vbet(2)(3); -- i va lua valoarea 77 

  vbet.EXTEND; -- se adauga un element de tip vector la vbet 

  vbet(5) := alfa(56,33); 

  vbet(4) := alfa(44,66,77,4321); 

  vbet(4)(4) := 7; -- 4321 este inlocuit cu 7 

  vbet(4).EXTEND; -- se adauga un element la al 4-lea element 

  vbet(4)(5) := 777; -- acest nou element adaugat va fi 777 

END; 

/ 

DECLARE 

  TYPE  gama IS TABLE OF VARCHAR2(20); 

  TYPE  delta IS TABLE OF gama; 

  TYPE  teta IS VARRAY(10) OF INTEGER; 

  TYPE  epsi IS TABLE OF teta;  

  var1  gama := gama('alb','negru'); 

  var2  delta := delta(var1); 

  var3  epsi := epsi(teta(31,15),teta(1,3,5)); 

BEGIN 

  var2.EXTEND; 

  var2(2) := var2(1); 

  var2.DELETE(1); -- sterge primul element din var2 

  /* sterge primul sir de caractere din al doilea  

     tablou al tabloului imbricat */ 

  var2(2).DELETE(1); 

END; 

/ 

DECLARE  

  TYPE  alfa IS TABLE OF INTEGER INDEX BY BINARY_INTEGER; 

  TYPE  beta IS TABLE OF alfa INDEX BY BINARY_INTEGER; 

  TYPE  gama IS VARRAY(10) OF VARCHAR2(30); 

  TYPE  delt IS TABLE OF gama INDEX BY BINARY_INTEGER; 

  var1  gama := gama('alb','negru'); 

  var2  beta; 

  var3  delt; 

  var4  alfa; 

  var5  alfa; -- tablou vid 

BEGIN 

  var4(1) := 324; 

  var4(2) := 222; 

  var4(42) := 333; 

  var2(27) := var4; 

  var3(39) := gama(77,76,89,908); 

  -- var2(40)(3) := 55; eroare nu exista element 40 in var2 

  var2(40) := var5; -- asignez un tablou null 

  var2(40)(3) := 55; -- corect 

END; 

/ 

Prelucrarea colecţiilor 

O colecţie poate fi exploatată fie în întregime (atomic) utilizând comenzi 
LMD, fie pot fi prelucrate elemente individuale dintr-o colecţie (piecewise 

updates) utilizând operatori SQL sau anumite facilităţi oferite de PL/SQL. 

 Comanda INSERT permite inserarea unei colecţii într-o linie a unui 

tabel. Colecţia trebuie să fie creată şi iniţializată anterior.  

 Comanda UPDATE este folosită pentru modificarea unei colecţii 

stocate. 

 Comanda DELETE poate şterge o linie ce conţine o colecţie.  

 Colecţiile din baza de date pot fi regăsite în variabile PL/SQL, 

utilizând comanda SELECT.  

Exemplu: 

CREATE OR REPLACE TYPE operalist AS VARRAY(10) OF 

                                    NUMBER(4); 

CREATE TABLE gal_ope ( 

  cod_galerie   NUMBER(10), 

  nume_galerie  VARCHAR2(20), 

  info          operalist); 

DECLARE  

  v_opera    operalist := operalist (777, 888, 999); 

  v_info_op  operalist := operalist (7007); 

  v_info     gal_ope.info%TYPE; 

  v_cod      gal_ope.cod_galerie%TYPE := 2345; 

BEGIN 

  INSERT INTO gal_ope 

  VALUES (4567, 'Impresionisti', operalist(4567,4987)); 

  INSERT INTO gal_ope 

  VALUES (2345, 'Cubism', v_opera); 

  INSERT INTO gal_ope 

  VALUES (123, 'Alfa', v_info_op); 

  SELECT info 

  INTO   v_info 

  FROM   gal_ope 

  WHERE  cod_galerie = v_cod; 

END; 

Un vector stocat într-un tabel este prelucrat ca un întreg (nu pot fi 

modificate elemente individuale). Prin urmare, elementele individuale ale unui 

vector nu pot fi referite în comenzile INSERT, UPDATE sau DELETE. Pentru 

referirea acestora trebuie utilizate comenzi procedurale PL/SQL. Pentru a 

modifica un vector, el trebuie selectat într-o variabilă PL/SQL a cărei valoare 

poate fi modificată şi apoi reinserată în tabel.  

Tablourile imbricate depuse în baza de date sunt mai flexibile, deoarece 

pot fi prelucrate fie în întregime, fie ca elemente individuale. În fiecare caz pot fi 

utilizate numai comenzi SQL.  

Se pot face reactualizări sau inserări asupra tablourilor imbricate care dau 

o valoare nouă pentru întreaga colecţie sau se pot face inserări, ştergeri, 

reactualizări de elemente particulare din colecţie. 

O colecţie poate fi asignată altei colecţii prin comenzile INSERT, 

UPDATE, FETCH, SELECT, instrucţiunea de atribuire sau prin apelul unui 

subprogram, dar colecţiile trebuie să fie de acelaşi tip. Dacă unei colecţii i se 

asignează o colecţie atomic null, aceasta devine atomic null şi trebuie 

reiniţializată. 

În Oracle8i a fost introdus operatorul TABLE, ce permite prelucrarea 

elementelor unui tablou imbricat care este stocat într-un tabel. Operatorul 

permite interogarea unei colecţii în clauza FROM (la fel ca un tabel).  

Operandul lui TABLE este:  

 fie numele unei colecţii şi atunci rezultatul operatorului este tot o 

colecţie, 

 fie este o subinterogare referitoare la o colecţie, iar în acest caz, 

operatorul TABLE returnează o singură valoare (coloană) care este un 

tablou imbricat sau un vector. Prin urmare, lista din clauza SELECT a 

subcererii trebuie să aibă un singur articol.  

Exemplu: 

Se presupune că tabelul opera are o coloană info de tip tablou imbricat. 

Acest tablou are două componente în care pentru fiecare operă de artă sunt 

depuse numele articolului referitor la opera respectivă şi revista în care a apărut. 

Să se insereze o linie în tabelul imbricat. 

INSERT INTO TABLE (SELECT info  

                   FROM   opera 

                   WHERE  titlu = 'Primavara') 

VALUES ('Pictura moderna', 'Orizonturi'); 

Listarea codului fiecărei opere de artă şi a colecţiei articolelor referitoare 

la aceste opere de artă se face prin comanda: 

SELECT  a.cod_opera, b.* 

FROM    opera a, TABLE (a.info) b; 

Pentru tablouri imbricate pe mai multe niveluri, operaţiile LMD pot fi 

făcute atomic sau pe elemente individuale, iar pentru vectori pe mai multe 

niveluri, operaţiile pot fi făcute numai atomic. 

Pentru prelucrarea unei colecţii locale se poate folosi şi operatorul CAST. 

CAST are forma sintactică: 

CAST (nume_colecţie AS tip_colecţie) 

Operanzii lui CAST sunt o colecţie declarată local (de exemplu, într-un 

bloc PL/SQL anonim) şi un tip colecţie SQL. CAST converteşte colecţia locală la 

tipul specificat. În felul acesta, o colecţie poate fi prelucrată ca şi cum ar fi un 

tabel SQL al bazei de date. 

Metodele unei colecţii 

PL/SQL oferă subprograme numite metode (methods), care operează 

asupra unei colecţii. Acestea pot fi apelate numai din comenzi procedurale, şi 

nu din SQL. 

Metodele sunt apelate prin expresia: 

nume_colecţie.nume_metodă [ (parametri) ] 

Metodele care se pot aplica colecţiilor PL/SQL sunt următoarele: 

COUNT returnează numărul curent de elemente ale unei colecţii PL/SQL; 

DELETE(n) şterge elementul n dintr-o colecţie PL/SQL; DELETE(m, n) 

şterge toate elementele având indecşii între m şi n; DELETE şterge 

toate elementele unei colecţii PL/SQL (nu este validă pentru tipul 

varrays); 

EXISTS(n) returnează TRUE dacă există al n-lea element al unei colecţii 

PL/SQL (altfel, returnează FALSE, chiar dacă elementul este null); 

FIRST, LAST returnează indicele primului, respectiv ultimului element 

din colecţie; 

NEXT(n), PRIOR(n) returnează indicele elementului următor, respectiv 

precedent celui de rang n din colecţie, iar dacă nu există un astfel de 

element returnează valoarea null; 

EXTEND adaugă elemente la sfârşitul unei colecţii: EXTEND adaugă un 

element null la sfârşitul colecţiei, EXTEND(n) adaugă n elemente null, 

EXTEND(n, i) adaugă n copii ale elementului de rang i (nu este validă 

pentru tipul index-by tables); nu poate fi utilizată pentru a iniţializa o 

colecţie atomic null; 

LIMIT returnează numărul maxim de elemente ale unei colecţii (cel de la 

declarare) pentru tipul vector şi null pentru tablouri imbricate (nu este 

validă pentru tipul index-by tables); 

TRIM şterge elementele de la sfârşitul unei colecţii: TRIM şterge ultimul 
element, TRIM(n) şterge ultimele n elemente (nu este validă pentru 
tipul index-by tables). Similar metodei EXTEND, metoda TRIM 
operează asupra dimensiunii interne a tabloului imbricat. 

EXISTS este singura metodă care poate fi aplicată unei colecţii atomice 

null. Orice altă metodă declanşează excepţia COLLECTION_IS_NULL.  

Bulk bind 

În exemplul care urmează, comanda DELETE este trimisă motorului SQL 

pentru fiecare iteraţie a comenzii FOR. 

Exemplu: 

DECLARE 

  TYPE nume IS VARRAY(20) OF NUMBER; 

  alfa nume := nume(10,20,70); -- coduri ale galeriilor 

BEGIN 

  FOR j IN alfa.FIRST..alfa.LAST 

    DELETE FROM opera 

    WHERE  cod_galerie = alfa (j); 

  END LOOP; 

END; 

Pentru a realiza mai rapid această operaţie, ar trebui să existe posibilitatea 

de a şterge (prelucra) întreaga colecţie şi nu elemente individuale. Tehnica care 

permite acest lucru este cunoscută sub numele bulk bind. 

În timpul compilării, compilatorul PL/SQL asociază identificatorii cu o 

adresă, un tip de date şi o valoare. Acest proces este numit binding. 

Comenzile SQL din blocurile PL/SQL sunt trimise motorului SQL pentru 

a fi executate. Motorul SQL poate trimite înapoi date motorului PL/SQL (de 

exemplu, ca rezultat al unei interogări). De multe ori, datele care trebuie 

manipulate aparţin unei colecţii, iar colecţia este iterată printr-un ciclu FOR. 

Prin urmare, transferul (în ambele sensuri) între SQL şi PL/SQL are loc pentru 

fiecare linie a colecţiei. 

Începând cu Oracle8i există posibilitatea ca toate liniile unei colecţii să 

fie transferate simultan printr-o singură operaţie. Procedeul este numit bulk bind 

şi este realizat cu ajutorul comenzii FORALL, ce poate fi folosită cu orice tip de 

colecţie. 

Comanda FORALL are sintaxa: 

FORALL index IN lim_inf..lim_sup 

  comanda_sql; 

Motorul SQL execută comanda_sql o singură dată pentru toate valorile 

indexului. Comanda_sql este una din comenzile INSERT, UPDATE, DELETE 

care referă elementele uneia sau mai multor colecţii. Variabila index poate fi 

referită numai în comanda FORALL şi numai ca indice de colecţie. 

În exemplul care urmează este optimizată problema anterioară, în sensul 

că instrucţiunea DELETE este trimisă motorului SQL o singură dată, pentru toate 

liniile colecţiei. 

Exemplu: 

DECLARE 

  TYPE nume IS VARRAY(20) OF NUMBER; 

  alfa nume := nume(10,20,70); -- coduri ale galeriilor 

BEGIN 

  … 

  FORALL j IN alfa.FIRST..alfa.LAST 

    DELETE FROM opera 

    WHERE  cod_galerie = alfa (j); 

END; 

Pentru utilizarea comenzii FORALL este necesară respectarea 

următoarelor restricţii: 

 comanda poate fi folosită numai în programe server-side, altfel apare 

eroarea “this feature is not supported in client-side programs”; 

 comenziile INSERT, UPDATE, DELETE trebuie să refere cel puţin o 

colecţie; 

 toate elementele colecţiei din domeniul precizat trebuie să existe 

(dacă, de exemplu, un element a fost şters, atunci este semnalată o 

eroare); 

 indicii colecţiilor nu pot să fie expresii şi trebuie să aibă valori 

continue. 

Exemplu: 

CREATE TABLE exemplu (x NUMBER, y NUMBER); 

DECLARE 

  TYPE  nume IS TABLE OF NUMBER; 

  ttt   nume:= nume(1,2,3); 

BEGIN 

  FORALL i IN ttt.FIRST..ttt.LAST 

    INSERT INTO exemplu 

    VALUES(ttt(i), 10);  -- corect 

  FORALL i IN 1..3 

    INSERT INTO exemplu 

    VALUES(7, 9);        -- exceptie 

END; 

 

 

  FORALL i IN gama.FIRST..gama.LAST 

    DELETE FROM carte 

    WHERE codel = gama(i+1);   

  -- eroare dubla (expresie si >LAST) 

DECLARE 

  TYPE alfa IS TABLE OF NUMBER; 

  xx   alfa := alfa(10,20,30,40); 

BEGIN 

  xx.DELETE(3); 

  FORALL i IN xx.FIRST..xx.LAST 

    DELETE FROM carte 

    WHERE codel = xx(i);  -- eroare 

END; 

Dacă există o eroare în procesarea unei linii printr-o operaţie LMD de tip 

bulk, numai acea linie va fi rollback. 

Cursorul SQL are un atribut compus %BULK_ROWCOUNT care numără 

liniile afectate de iteraţiile comenzii FORALL. %BULK_ROWCOUNT(i) 

reprezintă numărul de linii procesate de a i-a execuţie a comenzii SQL. Dacă nu 

este afectată nici o linie, valoarea atributului este 0. %BULK_ROWCOUNT nu 

poate să fie parametru în subprograme şi nu poate fi asignat altor colecţii. 

Începând cu Oracle9i este utilizabilă o nouă clauză în comanda FORALL. 

Clauza, numită SAVE EXCEPTIONS, permite ca toate excepţiile care apar în 

timpul execuţiei comenzii FORALL să fie salvate şi astfel procesarea poate să 

continue. În acest context, poate fi utilizat atributul cursor 

%BULK_EXCEPTIONS pentru a vizualiza informaţii despre aceste excepţii.  

Atributul acţionează ca un tablou PL/SQL şi are două câmpuri: 

 %BULK_EXCEPTIONS(i).ERROR_INDEX, reprezentând iteraţia în 

timpul căreia s-a declanşat excepţia; 

 %BULK_EXCEPTIONS(i).ERROR_CODE, reprezentând codul 

Oracle al erorii respective. 

Regăsirea rezultatului unei interogări în colecţii (înainte de a fi trimisă 

motorului PL/SQL) se poate obţine cu ajutorul clauzei BULK COLLECT.  

Clauza poate să apară în: 

 comenzile SELECT INTO (cursoare implicite),  

 comenzile FETCH INTO (cursoare explicite), 

 clauza RETURNING INTO a comenzilor INSERT, UPDATE, 

DELETE.  

Clauza are următoarea sintaxă:  

…BULK COLLECT INTO nume_colecţie [,nume_colecţie]… 

DECLARE 

  TYPE tip1 IS TABLE OF opera.cod_opera%TYPE; 

  TYPE tip2 IS TABLE OF opera.titlu%TYPE; 

  alfa  tip1; 

  beta  tip2; 

BEGIN 

  … 

  /* motorul SQL incarca in intregime coloanele 

     cod_opera si titlu in tabelele imbricate, 

     inainte de a returna  

     tabelele motorului PL/SQL */ 

  SELECT cod_opera, titlu BULK COLLECT INTO alfa,beta 

  FROM   opera; 

  … 

  /* daca exista n opere de arta in stare buna, 

     atunci alfa va contine codurile celor n opere */ 

  DELETE FROM opera WHERE stare = 'buna' 

  RETURNING cod_opera BULK COLLECT INTO alfa; 

  … 

END; 

Comanda FORALL se poate combina cu clauza BULK COLLECT. Totuşi, 
trebuie subliniat că ele nu pot fi folosite simultan în comanda SELECT. 

 Motorul SQL incarca toate liniile unei coloane. Cum se poate limita 
numarul de linii procesate? Exemplelele urmatoare dau 2 variante de rezolvare. 

DECLARE 

  TYPE alfa IS TABLE OF carte.pret%TYPE; 

  xx   alfa; 

BEGIN 

  SELECT pret BULK COLLECT INTO xx 

  FROM   carte WHERE  ROWNUM <= 500; 

END; 

 

DECLARE 

  … 

  zz NATURAL := 10; 

BEGIN 

  OPEN c1; 

  LOOP 

    FETCH c1 BULK COLLECT INTO xx, yy LIMIT zz; 

    EXIT WHEN c1%NOTFOUND; 

    … 

END; 


